
Introduction

Speech is becoming more important as compute power increases and research contributes solutions that work. There are two aspects to speech that is of interest to application developers: speech generation and speech understanding. Speech generation is the easy half and has been available for quite some time. Speech understanding is substantially more difficult but it is making inroads. In this column, we’ll make use of speech generation. In the next column, we’ll try speech understanding. 



More specifically, we will implement an IBM VisualAge application that makes use of text to speech software called Monologue from First Byte that came bundled with a Sound Blaster 16 card. The application we have in mind is a computer substitute for reading bedtime stories to your kids. The application (see Figure 1) contains a file menu entry that can be used to open a file which is displayed in a text pane. To control the computerized reading of the file contents, three buttons are provided: start, restart, and stop.

Development Preamble

When we first developed this application, we built a special pool dictionary containing the names of the external C DLL functions associated with the text to speech software. One of the functions provided, for example, was “Say”. We called this pool dictionary FirstByteMonologueFunctions and initialized the “Say” entry as follows:



	FirstByteMonologueFunctions 

		at: #'Say' 

		put: (PlatformFunction 

			callingConvention: #pascal16

   			function: 'Say'

  			library: 'FB_SPCH.DLL'

   			parameterTypes: #(#int32 #pointer)

   			returnType: #int16)



Subsequently, it was possible to call this function in a method whose class referenced the pool dictionary as follows:



		Say 

			callWith: parameter1 

			with: parameter2 

			         …



Unfortunately, the approach turned out to be totally unfeasible when we started work on the speech recognition component because there were simply too many functions to tap into. We had to automate the process. So we build a “setup” application that could be used to create a FirstByteMonologuePackage and an IBMVoiceTypePackage that would contain the API specific dictionaries. These packages could then serve as prerequisites for our speech generation and recognition packages.

The Speech Setup Package

We designed the speech setup package to contain one application, SetupSpeech, which we created as a visual part using the composition editor. It is really just an elaborate big button application (actually two). The “setup speaker package” button (see Figure 2) creates FirstByteMonologuePackage, which we make use of in this column; the “setup listener package” button creates IBMVoiceTypePackage, which we intend to use in the next Smalltalk column. The SetupSpeech application also contains a progress bar (which isn’t really evident in the figure) and a status bar. To make use of these widgets, we had to load “Windows 95 Controls”. We also loaded “Language Interface, C” to make use of the facility, AbtCLangParser, that can be used to extract information about the API and C structures described in “*.h” files.



Although the application itself is trivial, a fair amount of code (see Listing 1) is needed to automatically generate the two packages. We set up the application so that the two push button #clicked events would trigger the corresponding methods setupSpeakerPackage and setupListenerPackage. These both make use of a generic method that goes through about a dozen steps. The step method is used to indicate the state of the installation in the progress bar. An appropriate status line (one of the sentences in a collection initialized by setupSteps) is also displayed in the status bar at the bottom of the application.



In summary, our setup application proceeds as follows. Step 1 (method setupInstallation) sets up the window title, the progress bar, and the status bar. Step 2 (method previouslySetup) determines if the package was previously set up (if it was, the user must unload the package from his image). 



Step 3 (method includeFileFound) prompts the user for a C “.h” file provided by the implementers of the DLL that we wish to connect to (either “monologue.h” or “smapi.h”, depending on which button was clicked). This file (and any other “.h” files that might potentially be included by this initial file) is copied into the Smalltalk working directory (apparently, the C language parser wasn’t designed to work from an arbitrary directory). 



Step 4 (method parseIncludeFile) invokes the C language parser on the above mentioned file which creates a table (we called it “nameTable) describing the C structures and function used in the “.h” file. 



Step 5 (method buildPackage) actually creates the package programmatically. 



Step 6 (method processIncludeFile) extracts the useful information from the name table, stores it into special variables “constants” and “functions”, and creates special Smalltalk classes (OS classes that can read/write information to/from C structures). As it turns out, there aren’t any generated for the speaker package, but a whole host are defined for the listener package. 



Step 7 (method buildMethods) creates and compiles a method in the package called loaded (which, in turn, uses two helper methods call loadConstants and loadFunctions) and another method called removing. Method loaded creates and initializes the pool dictionaries for the package and method removing removes the pool dictionary names from the global Smalltalk dictionary. For the speaker package, these pool dictionaries are respectively named FirstByteMonologueConstants  and FirstByteMonologueFunctions.  Creating the methods was actually quite easy. We built one string that contained all the method source (each method was delimited with “!”) and asked class EmFileOutInterface to file it in. The store string for the two pool dictionaries were plugged into the appropriate spots in the loadConstants and loadFunctions methods using a special built-in string message called bindWithArguments:. It’s a simple message to use



	'	Smalltalk at: #%1 put: (%2) !'

		bindWithArguments: (Array 

			with: 'Test' with: 'nil')



is equivalent to writing



	'	Smalltalk at: #Test put: (nil) !'



Step 8 (method initializePackage) executes the newly generated method “loaded” to create the two dictionaries (which is what actually happens when the packaged is loaded; we don’t have any choice as to what to call this method). It also attempts to version the methods and the package.



Step 9 (methods removeIncludeFiles) deletes the include files previously added to the Smalltalk directory in step 3. 



Step 10 (methods wrapupInstallation) resets the progress bar and adds a “done” message to the window title. 

The Speaker Package

We designed a class called SpeakerEngine (see listing 2), which we placed in a package called SpeakerPackage, which uses FirstByteMonologuePackage as a prerequisite. A user wishing to make use of speech generation has to make an instance of the engine (SpeakerEngine new), tell it to open (anEngine open), make it say one or more statements (anEngine say: aString), and then finally close it when the engine is no longer needed (anEngine close).

The StorybookReader Package

We finally designed our goal application called StorybookReader (see listing 2) which we placed in a package called StorybookReaderPackage. Just like the setup application, it was created visually using the composition editor. 



When the storybook reader is opened, an event (#aboutToOpenWidget) is triggered which causes method openReader to execute. This creates an instance of a speaker engine and opens it. When the storybook reader is closed, a corresponding (#closedWidget) event triggers the closeReader method which closes the engine.



As you can see from Figure 1, the storybook reader has a file menu bar entry with an “Open …” submenu. The #clicked event for this item triggers the handler (method openFile:) which prompts the user for a text file. The contents of the file is placed in the text widget and a stream (storyStream) is created to permit later extraction of the text one sentence at a time. Corresponding #clicked handlers (start, restart, and stop) for the buttons initiate or terminate the actual speech generation.



It is method start that actually contains the loop which extracts one sentence at a time and asks the engine to output it. Extracting a sentence is actually done by readABit. The readABit method additionally uses the starting and end positions in the stream to select the spoken text in the text widget (via methods markStart and markEnd). The reason we perform sentence extraction is that the speaker engine does not return control until after it has finished voicing the entire text. There are a number of facilities newly introduced into VisualAge that supposedly permits asynchronous execution of the C APIs. We tried them all and they all hung our application. 



Method readABit was initially written with a traditional loop that executed as long as the boolean reading was true. But we found that the Stop button would not react to our clicks until after the loop terminated. Conceptually, it’s easy to understand why. Method readABit is actually executing in response to a #clicked event triggered from either the Start or Restart buttons. The Stop button’s event handler can’t generate a response to its own #clicked event until after the first event handler is finished. In fact, the button won’t even depress until the first handler terminates. To solve this problem, we placed the loop in a block and forked a new process to execute it. In Smalltalk, it’s very easy to do. Simply execute



	[… new process code …] fork



This way, the Start (or Restart) #clicked event handler finishes right away allowing subsequent #clicked events for Stop to be handled. However, this won’t happen unless the newly forked process relinquishes control periodically. So we added the following to the loop.



	Processor yield



This is adequate but it’s still not quite perfect. The Stop button will still not depress while in the middle of a sentence. 

Conclusions

We built a very simple application to deal with speech generation. For the text to speech facility we used, all we needed was an open, say, and close facility. More sophisticated facilities with natural sounding speech generating capabilities are available. Presumably, they would be slightly more complex to control but hopefully, not a whole lot more complex.



When we tested our storybook reader the first time, we opened it on the first file that happened to be a text file. This turned out to be an IBM copyright file. After a few sentences of computer generated speech, we clicked on Stop but nothing happened. Even the Smalltalk Cntl-Break key sequence wouldn’t work. The only way to stop it was to kill the entire Smalltalk session. We didn’t want to do that because some of the changes made in the composition editor hadn’t be saved. We walked around the office a few times while the computer droned on and on. Ultimately, we took a short lunch break. 

Figure 1: Building the Storybook Reader.

Figure 2: The Setup Application.



Listing 1 The SetupSpeech class.



class:									SetupSpeech

superclass:							AbtAppBldrView 

instance variables:			packagePrefix package domainName 											includeFileName includeFileNames 

											nameTable dllName dllType constants 

											functions statusLines stepIndex



comment							A visual part



instance methods



setup facilities



setupSpeakerPackage

	self showBusyWhile: [

		self 

			setupPackage: #FirstByteMonologue 

			domain: #speaker 

			includeFile: 'monologue.h' 

			dll: 'FB_SPCH'

			type: #'16BitPascal']



setupListenerPackage

	self showBusyCursorWhile: [

		self 

			setupPackage: #IBMVoiceType

			domain: #listener 

			includeFile: 'smapi.h' 

			dll: 'SMAPI'

			type: #'32BitStandard']



setupPackage: packageNamePrefix 

domain: packageDomainName 

includeFile: fileName 

dll: dllFileName 

type: dllFileType

	

	packagePrefix := packageNamePrefix.

	domainName := packageDomainName.

	includeFileName := fileName.

	nameTable := nil. "to be filled in later"

	dllName := dllFileName.

	dllType := dllFileType.

	constants := nil. "to be filled in later" 

	functions := nil. "to be filled in later" 



	self setupInstallation.

	self step. self previouslySetup 

		ifTrue: [^self abort: 'Package previously setup'].

	self step. self includeFileFound

		ifFalse: [^self abort: 'Include file not found'].

	self step; parseIncludeFile.

	self step; buildPackage.

	self step; processIncludeFile.

	self step; buildMethods.

	self step; initializePackage.

	self step; removeIncludeFiles.

	self step; wrapupInstallation

	

setup steps



step

	self progressBar

		value: (stepIndex := stepIndex + 1).

	self inform: (statusLines at: stepIndex).

	(Delay forSeconds: 1) wait



setupInstallation

	self window title: 'Installing ', self capitalizedDomainName.

	self setupSteps; setupProgressBar.



setupSteps

	| replacements |

	stepIndex := 0.

	replacements := Array 

		with: domainName with: includeFileName.

	statusLines := #(	

		'Checking if %1 package already exists'

		'Requesting location of file ''%2'''

		'Processing file ''%2'''

		'Building DLL API functions'

		'Building %1 package'

		'Writing %1 package methods'

		'Initializing %1 package'

		'Final cleanup'

		'All done') collect: [:line |

			line bindWithArguments: replacements].



	



setupProgressBar

	self progressBar

		minimum: 1;

		maximum: statusLines size



previouslySetup

	^Smalltalk includesKey: self packageName



includeFileFound

	| name |

	"The include file can include others."

	name := self getIncludeDirectoryName.

	name isEmpty ifTrue: [^false].

	includeFileNames := self getIncludeFileNames: name.

	includeFileNames do: [:fileName |

		CfsFileDescriptor copy: name, fileName new: fileName].

	^true



parseIncludeFile

	nameTable := AbtCLangParser parseFile: includeFileName.

	dllType == #'16BitPascal'

		ifTrue: [nameTable make16Bit; makePascal]



buildPackage

	Application 

		create: self packageName 

		with: (Array with: Kernel).

	package := Smalltalk at: self packageName



processIncludeFile

	| className |

	constants := nameTable constants.

	functions := nameTable platformFunctions: dllName.

	nameTable dataStructures 

		keysAndValuesDo: [:name :structure |

			className := name asSymbol.

			OSStructure 

				subclass: className

				instanceVariableNames: ''

				classVariableNames: ''

				poolDictionaries: ''.



			AbtRecordGenerator new

				generateAllFieldsFrom: structure

				inClass: (Smalltalk at: className)]



buildMethods

	package becomeDefault.

	EmFileOutInterface 

		fileInSourceFrom: (ReadStream 

			on: self fileInString).



initializePackage

	package loaded.

	self versionAll



removeIncludeFiles

	includeFileNames do: [:fileName |

		CfsFileDescriptor remove: fileName]



wrapupInstallation

	self wrapupProgressBar.

	self window 

		title: 'Done installing ', self capitalizedDomainName.



wrapupProgressBar

	self progressBar 

		value: self progressBar minimum



setup steps support



packageName

	^(packagePrefix, 'Package') asSymbol



capitalizedDomainName

	^(domainName asString copy

		at: 1 put: domainName first asUppercase;

		yourself), ' Package'



fileInString

	| replacements lines |

	replacements := Array

		with: packagePrefix

		with: constants storeString

		with: functions storeString.

	lines := #(

		'!%1Package class publicMethods !'

		''

		'loadConstants'

		'	Smalltalk at: #%1Constants put: (%2) !'

		''

		'loadFunctions'

		'	Smalltalk at: #%1Functions put: (%3) !'

		''

		'loaded'

		'	self loadConstants; loadFunctions !!'

		''

		'removing'

		'	Smalltalk removeKey: #%1Constants.'

		'	Smalltalk removeKey: #%1Functions !! !!').



	^(lines inject: '' into: [:string :line |

		string, CldtConstants::LineDelimiter, line])

			bindWithArguments: replacements.



getIncludeDirectoryName

	| pathName suffix |

	pathName := self prompter 

		title: 'Please locate file ''', includeFileName, '''';

		prompt;

		selectedFileName.

	pathName isEmpty ifTrue: [^''].

	suffix := pathName abrSimpleFilename.

	^pathName copyFrom: 1 to: pathName size - suffix size.



getIncludeFileNames: directoryName

	| fileList names entry fileName |

	fileList := CfsDirectoryDescriptor 

		opendir: directoryName 

		pattern: '*.h' 

		mode: CfsConstants::FREG.

	names := OrderedCollection new.

	[(entry := fileList readdir) notNil] whileTrue: [

		fileName := entry dName.

		names add: fileName].

	^names



inform: message

	self statusBar

		labelString: message



abort: note

	| message |

	message := 'Aborted ... ', note.

	self inform: message.

	CwMessagePrompter errorMessage: message.

	self window title: 'Aborted ', 

		self capitalizedDomainName, ' Installation'.

	self wrapupProgressBar



showBusyWhile: aBlock

	"Fork to ensure the status bar updates.

	Also requires disabling to prevent

	additional reactions to clicks while

	not done."

	[

		self window disable.

		System showBusyCursorWhile: [

			aBlock value.

		self window enable]

	] fork.



versionAll

	| classes |

	classes := package defined select: [:class |

		self versionClass: class].

	package releaseEachClassIn: classes.

	(classes size = package defined size)

		ifTrue: [

			self versionPackage ifFalse: [

				System message: 'Package unsuccessfully versioned']]

		ifFalse: [

			System message: 'Not all classes successfully versioned']



versionClass: class

	"Returns a boolean."

	| name |

	name := EmInterface current 

		availableClassVersionNameFor: class 

		in: package.

	(name isNil or: [name isEmpty]) ifTrue: [^false].

	^class versionNameIn: package is: name



versionPackage

	"Returns a boolean."

	| name |

	name := EmInterface current  

		availableApplicationVersionNameFor: package.

	(name isNil or: [name isEmpty]) ifTrue: [^false].

	^package versionName: name



widget references



progressBar

	^self subpartNamed: 'Progress Bar'



prompter

	^self subpartNamed: 'File Selection Prompter'



statusBar

	^self subpartNamed: 'Status Bar'

	

window

	^self subpartNamed: 'Window'





Listing 2 The SpeakerEngine class.



class:									SpeakerEngine

superclass:							Object

instance variables:			soundControlBlock 

pool dictionaries:				FirstByteMonologueFunctions



comment							A non-visual part.



class methods 



examples



example1

	"SpeakerEngine example1"

	| sentences engine |

	engine := SpeakerEngine new.

	sentences := #(

		'A good morning omelette'

		'4 large eggs' 

		'a pinch of salt'

		'a little milk'

		'stir and cook over medium heat').

	engine open.

	sentences do: [:sentence | engine say: sentence].

	engine close



instance methods



registering/deregistering



open 

	soundControlBlock := 

		OpenSpeech callWith: 0 with: 0 with: 0.



close 

	CloseSpeech callWith: soundControlBlock



saying



say: sentence 

	Say callWith: soundControlBlock with: sentence asPSZ



Listing 3 The StorybookReader class.



class:									StorybookReader

superclass:							AbtAppBldrView 

instance variables:			storyStream speakerEngine reading



comment							A visual part.



instance  methods 



part referencing



restartButton

	^(self subpartNamed: 'Restart') 

startButton

	^(self subpartNamed: 'Start') 

stopButton

	^(self subpartNamed: 'Stop') 

textPart

	^(self subpartNamed: 'Text Part') 



startup/wrapup event handling



openReader

	self prepareForFile.

	speakerEngine := SpeakerEngine new.

	speakerEngine open



closeReader

	speakerEngine close



opening



openFile: fileName

	| fileStream everything |

	fileStream := CfsReadFileStream open: fileName.

	everything := fileStream contents.

	fileStream close.

	storyStream := ReadStream on: everything.

	self textPart string: everything.

	self prepareForStarting



button event handling



start

	[self reading: true.

	self prepareForStopping.

	[self readingInProgress] whileTrue: [

		self readABit.

		Processor yield].

	self prepareForStarting] fork



restart

	storyStream reset.

	self start



stop

	self reading: false.



button enabling/disabling



prepareForFile

	self startButton disable.

	self restartButton disable.

	self stopButton disable.



prepareForStarting

	self startButton enabled: self readingCanContinue.

	self restartButton enable.

	self stopButton disable.



prepareForStopping

	self startButton disable.

	self restartButton disable.

	self stopButton enable.



text reading



reading

	^reading 



reading: aBoolean

	reading := aBoolean



readingCanContinue

	^storyStream atEnd not



readingInProgress

	^self reading and: [self readingCanContinue]



specialSuffixEncountered: candidate

	| tail |

	#('Mr.' 'Mrs.' 'Ms.' 'Dr.' 'Sr.' 'Jr.') do: [:suffix |

		candidate size >= suffix size ifTrue: [

			tail := candidate 

				copyFrom: candidate size - suffix size + 1 

				to: candidate size.

			suffix = tail ifTrue: [^true]]].

	^false



readABit

	| sentence |

	self markStart.

	sentence := self nextSentenceToRead.

	self markEnd.

	speakerEngine say: sentence



nextSentenceToRead

	| candidate |

	candidate := storyStream abrUpToAny: #($. $!! $?).

	storyStream atEnd ifFalse: [

		storyStream skip: -1. 

		candidate := candidate copyWith: storyStream next].

	(self longEnoughCandidateSentence: candidate)

		ifTrue: [^candidate]

		ifFalse: [^candidate, self nextSentenceToRead]



longEnoughCandidateSentence: candidate

	storyStream atEnd ifTrue: [^true].

	storyStream peek isSeparator not ifTrue: [^false].

	candidate size < 20 ifTrue: [^false].

	(self specialSuffixEncountered: candidate) ifTrue: [^false].

	^true!





text scrolling



markStart

	self textPart cursorPosition: storyStream position



markEnd

	self textPart

		selectTextFrom: self textPart cursorPosition

		to: storyStream position

