
Introduction

Previously, we developed a speaker engine that provided an interface to text to speech software called Monologue from First Byte (this software came bundled with a Sound Blaster 16 card). We also implemented a setup package that generated the required C interface functions and constants from the “.h” files supplied with the software. We subsequently used the speaker engine to develop a storybook reader that could read an arbitrary text file out loud.



Now, we want to develop a similar engine for speech recognition using a small subset of IBM VoiceType. To differentiate from the other engine, which was called SpeakerEngine, we will call this one ListenerEngine. This engine will serve as a building block for our real goal which we will discuss in a follow-up column. What we really want to developed is a voice-controlled recipe browser (as shown in Figure 1) that is intended to be used in the kitchen. We want users to be able to select a recipe file and a recipe within that file with the mouse. However, from this point onward, the user to be able to switch to voice control as the actual recipes are being followed. Once your hands are “wet” with cooking ingredients, it’s not possible to work the mouse. By default, the browser will be ready to list the ingredients via commands such as “next” or “previous”. It will also support switching to the list of instructions with command “instructions” or “steps”. Commands “next” and “previous” will apply to the instructions until the user switches back to ingredients via command “ingredients” or “stuff.” 



As it turns out, we can’t do the whole thing in one column. So we’ll tackle the listener engine in this column and the actual recipe browser in the next one.

Using the Listener Engine

Before we discuss the implementation of the listener engine, it makes sense to have a good understanding of how it is intended to be used. This way, it will be easier to distinguish the important details from the incidental. To illustrate its intended use, we built a very simple demonstration class, ListenerDemonstration (see Listing 1). 



Recall that the speaker engine that we designed in the previous column understood only three messages: open, “say: aString”, and close. We would like the listener engine to emulate this simplicity as much as possible. So, to begin, we need an open and close. The equivalent to saying something is to listen for something. But there is a subtle difference. When you say something, you say it right away. When you listen, by contrast, you might listen for a second, a minute, three minutes, … Your software can’t be hung just listening for something. It has to be event oriented; i.e., when the engine hears something, it should report it to you when it hears it. We need a message like listen, which really means “start listening” and perhaps the converse, ignore, which really means “stop listening.” This should be it adequate for the kinds of simple applications we have in mind. However, some speech recognition systems need a bit more help. In the case of IBM VoiceType, in particular, it requires at least a preliminary list of the words it might be expected to hear. So we added one more message, “expect: stringCollection” (a word list).



Because our listening engine is event-oriented, it is provided with a client (any object) that it reports to when the events occur. In keeping with our goal of simplicity, we provide only four events: #opened, #closed, #heard:, and #distracted. Event #opened is triggered when the engine is ready to go. Event #closed is triggered after the user has sent close to the engine and it has successfully terminated. Event #heard: is triggered whenever a word is heard. The word that was heard (a string) is passed along as a parameter. Finally, #distracted is triggered when the underlying recognition engine is taken over by some other application. This means that it is no longer listening for you (it is listening for the other application instead). This will happen, for example, when the other application gains focus. When your application is reactivated, the listener engine can be reactivated by being told to listen again.



In the ListenerDemonstration class (Listing 1), you will note that method run obtains a listener engine by executing “ListenerEngine on: self”; self here is the client of the engine, thus all event messages will be sent to this client. The four event handlers (methods reactToOpen, reactToHeard:, reactToDistracted, and reactToClosed) are specified with messages of the following form:



	aListenerEngine

		when: #eventName 

		send: #messageSelector



The listener engine also supports sending messages to other objects using the message when:send:to:. In this demonstration, we create a workspace. When the engine is ready, event #opened is triggered causing reactToOpened to execute. This method outputs the words “hide, show, close, stop, quit, and exit” in the workspace, tells the engine to expect these words (the reject: message simply gets rid of the commas), and asks the engine to listen. When the engine hears a word, event #heard: is triggered causing method reactToHeard: to execute with the word that was heard.  As you can see from the method, we can make the workspace hide, show, and close. When noise is heard, the listener engine returns the word ''. This could be used as a visual cue that something was heard even though it was undecipherable. When event #distracted is triggered, method reactToDistracted simply closes the engine. So running a second example causes the first to close. Finally, closing the engine causes the #closed event to be triggered, which causes method reactToClosed to execute, which, in turn, closes the window.



Now that we have an understanding of the listening engine’s design, we are in a better position to consider its implementation. Three aspects need to be discussed: (1) the construction of a package that we called  IBMVoiceTypePackage containing interface functions to IBM’s speech recognition software, (2) a special class, OSListenerShell, that permits us to implement our simple design for the listener engine, and finally, (3) the listener engine itself.

The IBM VoiceType Package

After installing IBM VoiceType in a suitable directory, we used the setup application discussed in the previous column to create the required C interface functions and constants in a package called IBMVoiceTypePackage. All we had to do here was click on the “setup listener package” button. The setup application prompted us for the location of the file “smapi.h” (IBM loves acronyms) and did everything else. It actually generated 13 C structure classes including SmArg, SmVocword, and SmWord (the only three we actually used). Just to give you a flavor of the methods generated with these classes, consider the following method generated in class SmVocword:



	spelling

 		"Return the value of field spelling " 

		^ self abtPointerAt: 8 



This method retrieves a pointer which in Smalltalk appears as an integer. Although it is not at all obvious, we can retrieve the string pointed to by such a pointer using



	String 

		copyFromOSMemory: 

			anSmVocword spelling



We did have some problems with these methods. The C language parser that generated these methods assumes byte alignment for all types. As a consequence, C code compiled with double word alignment, for example, will not match up with the corresponding declarations in Smalltalk. To be more specific, the variables in the structure



		struct {

			long a;

			short b;

			long c;

		}



when byte aligned are at offsets 0 (4 bytes), 4 (2 bytes), 6 (4 bytes) respectively, for a total of 10 bytes. When aligned on 4 byte boundaries, the offsets are 0 (4 bytes), 4 (2 bytes + 2 bytes of padding), 8 (4 bytes) for a total of 12 bytes. We had to change 2 generated classes manually to make things work (the details were recorded in the package class’ comment).



It was now a simple matter to create a listener package with the above package as a prerequisite and to add the two classes OSListenerShell and ListenerEngine to the package. 

The OSListenerShell Class

IBM VoiceType was designed to work with windows events by sending special wmCommand messages to an active window. One possible design option was to revise the basic wmCommand handler in VisualAge to accommodate these new messages. However, this would still require Smalltalk users to understand how windows events are made available to their application in order to use the speech recognition facilities. As we alluded to above, it would be nicer to actually hide this complicated windows event layer by providing a listening engine that can send simple messages to any object whatsoever (not just to a window). A better design can be obtained by making the engine itself an invisible window. That way, we can design a very simple window to deal with the low-level windows events and have it pass the buck in a natural way to any object designed to react to speech recognition directives. There is no need to have all windows understand speech events.



To be able to satisfy our design goals, it is enough to introduce one new kind of window that understands speech commands. We called this new class OSListenerShell (see Listing 2) since it needs to inherit from OSShell. The class contains a windows level instance method wmCommand:with: that specifically looks for a command that we arbitrarily numbered WmUser + 10 (this magic number can be retrieved using message commandIdentifier). When the wParam is equal to the magic number, the lParam is the speech recognition software’s event handle. The listener engine (see Listing 3) was designed to be a top-level shell that opens invisibly. When an instance of ListenerEngine is obtained via code such as



	ListenerEngine on: anObject



the listener engine (itself an application shell) is set up (see class method on: in ListenerEngine) to be invisible when opened. In this Smalltalk system, a window is invisible when it is unmapped; visible otherwise (this is Motif terminology). When an open message is sent to the listener by the user (see method open), message realizeWidget actually opens the window (invisibly). Method close, the converse, executes destroyWidget to terminate it. When realizeWidget executes, message osCreateWidget is sent to the listener. This method which exists in the superclass was copied down to the listener engine class and changed to created an OSListenerShell instead of an OSShell. For non-IBM Smalltalk fans, the shell is the top pane or main view. An alternative to copying the entire method would have been to change it to say “self shellClass” permitting us to redefine a much simpler method such as the following:



	shellClass

		^OSListenerShell 



This explains how our listener engine is coupled to the OSListenerShell but doesn’t yet explain how the IBM VoiceType software knows to send messages with the above magic number. The connection is made in method callOpen which provides “self osWidget commandIdentifier” as an argument to a “session open” API call.



Now let’s consider the engine in more detail.

The ListenerEngine Class



The ListenerEngine (see Listing 3) is designed to interface with IBM VoiceType through an invisible window. It makes use of many VoiceType function calls whose names are contained by the pool dictionaries IBMVoiceTypeConstants and IBMVoiceTypeFunctions. A direct call to a C function appears as follows:



	SmSesMicOn 

		callWith: … parameter 1 …

		with: … parameter 2 …

		with: … parameter 3 …



All such functions typically return error codes, many of which are benign (for example, responding with “OK” or “the microphone is already on”). But just in case there is a serious error, we want to check this error code. However, we don’t want to perform the test pervasively. So we wrote an interface method that executes the function and also performs the test allowing us to write the following instead:



	self

		call: SmSesMicOn 

		with: … parameter 1 …

		with: … parameter 2 …

		with: … parameter 3 …



Only the one-parameter version is shown in the listing. Additionally, many functions return their values indirectly in an integer whose address is passed to the function. Such a memory location can be obtained via method newResultContainer



	newResultContainer

		^OSInt32 new 

				abtMoveToOSMemory



Passing this object as a parameter is equivalent to passing the address of the memory location. To retrieve the result, we can use the following method:



	contentsOf: container

		^container uint32At: 0 



Since every call requires at least one of these “container” objects, we created an instance variable “resultContainer” to contain one permanently.  



Most of the function calls require only simple parameters. But a few, illustrated for example by methods callOpen and callConnect, require the address of an SmArg structure which contains additional SmArg instances with a name (a key such as SmNapplicationName) and a value (such as a string). To deal with this, we defined a special method convertArgumentDictionary: which is too cumbersome to explain in detail. In fact, because the listener engine is primarily an interface to a C library, there are so many of these low-level issues to contend with that it’s probably not a good idea to focus on them.



So let’s consider the engine from a slightly higher level perspective. First, it maintains a client (the object that is intended to receive event handling messages) and a handler list (a dictionary) that map the event names #opened, #closed, #heard:, and #distracted to the client’s selectors. These client selectors are recorded by methods when:send: and when:send:to:. The engine also maintains a session number which is obtained by method callOpen, invoked when the user sends the message open. Method open also invokes callConnect to make the session active. 



The heart of the processing for the listener engine begins in the OSListenerShell’s instance method wmCommand:with: method in Listing 2. When the shell receives a speech event, it is forwarded to our engine using message processSpeechMessage:. The corresponding method (in Listing 3) decodes the low-level “lParam”. If it is one of the special IBM VoiceType messages (see method engineMessage for a list that includes, for example, SmRecognizedWord), it is translated to the corresponding user event name and dispatched to the client.



The only other interesting methods are the group of public methods open, close, “expect: words”, listen, and ignore under method category “user interaction”. Most of these are relatively obvious. Method listen, for example, requests the focus, turns the microphone on, and instructs the engine to anticipate recognizing a word. Each of these steps is implemented as a simple method; requestFocus, microphoneOn, and recognizedNextWord respectively. 

Conclusions

We built a very simple Smalltalk engine for speech recognition that hides and uses the relatively sophisticated facilities provided by IBM VoiceType. It provides the user with five messages (open, close, “expect: words”, listen, and ignore) and four events (#opened, #closed, #heard:, and #distracted) that can be directed to an arbitrary object. We hope to use this listener engine to built a recipe browser in the next column.

Figure 1: The Voice Controlled Recipe Browser.



Listing 1 The ListenerDemonstration class.



class:									ListenerDemonstration

superclass:							Object 

instance variables:			engine window 

class methods

examples

example1

	"ListenerDemonstration example1"

	ListenerDemonstration new run



instance methods

running

run

	window := EtWorkspace new open.

	engine := ListenerEngine on: self.

	engine

		when: #opened send: #reactToOpened;

		when: #heard: send: #reactToHeard:;

		when: #distracted send: #reactToDistracted;

		when: #closed send: #reactToClosed;

		open



reactToOpened

	| words |

	words := 'hide, show, close, quit, exit'.

	self show: 'Use words from list: ', words, '.'. 

	engine 

		expect: (words reject: [:character | 

			character == $,]) subStrings;

		listen



reactToClosed

	window close.



reactToHeard: word

	(#('close' 'quit' 'exit') includes: word) ifTrue: [^engine close].

	word = 'hide' ifTrue: [window shell iconify] ifFalse: [

	word = 'show' ifTrue: [window bringToFront] ifFalse: [

	word = '' ifTrue: [self show: 'Heard noise'] ifFalse: [

	"otherwise" self show: 'Heard ', word, '???']]].

	engine listen "for next time"



reactToDistracted

	self show: 'Something distracted me, bye!'.

	(Delay forSeconds: 5) wait.

	engine close



show: message

	window bringToFront; cr; tab; show: message



Listing 2 The OSListenerShell class.



class:									OSListenerShell

superclass:							OSShell 

instance variables:			"none"

class variables: 				CommandIdentifier

pool dictionaries:				PlatformConstants 



class methods

initializing

initialize

	"OSListenerShell initialize"

	CommandIdentifier := WmUser + 10



instance methods

querying

commandIdentifier

	^CommandIdentifier



event handling

wmCommand: wParam with: lParam

	"Process messages intended for the speech engine specially."

	wParam = CommandIdentifier ifTrue: [

		owner processSpeechMessage: lParam. ^nil].

	^super wmCommand: wParam with: lParam





Listing 3 The ListenerEngine class.



class:									ListenerEngine

superclass:							CwTopLevelShell 

instance variables:			client handlers sessionNumber 

											resultContainer isOpen

pool dictionaries:				IBMVoiceTypeConstants 

											IBMVoiceTypeFunctions



class methods

instance creation

on: client

	^self 

		createApplicationShell: 'Speech Shell' 

		argBlock: [:shell | 

			shell 

				mappedWhenManaged: false;

				client: client]



instance methods

user interaction

open

	self realizeWidget; callOpen; callConnect



close

	| closeHandlers |

	self client: nil. 

	closeHandlers := handlers at: #closed ifAbsent: [#()].

	handlers := Dictionary new.

	self call: SmSesClose with: self sessionNumber.

	self destroyWidget.

	isOpen := false.

	self userDispatchHandlers: closeHandlers with: #()



expect: words

	self defineVocabulary: 'expectations' words: words.

	self enableVocabulary: 'expectations'



listen

	^self requestFocus; microphoneOn; recognizeNextWord

ignore

	^self haltRecognizer; microphoneOff; releaseFocus



when: messageIdentifier send: selector

	self when: messageIdentifier send: selector to: self client



when: messageIdentifier send: selector to: object

	(handlers 

		at: messageIdentifier 

		ifAbsentPut: [OrderedCollection new: 1])

			add: (DirectedMessage new 

				receiver: object; selector: selector)

		

private initializing

initialize

	super initialize.

	resultContainer := self newResultContainer.

	handlers := Dictionary new.

	isOpen := false



private overriding

osCreateWidget

	"Overridding: changed OSShell to OSListenerShell."

				… code not shown …



private interfacing support

call: function with: parameter1

	self for: function check: (function callWith: parameter1)



… other variations with more with: keywords …



for: function check: responseCode

	responseCode = SmRcOk ifTrue: [^self].

	responseCode = SmRcMicAlreadyOn ifTrue: [^self].

	responseCode = SmRcMicAlreadyOff ifTrue: [^self].

	responseCode = SmRcBadAp ifTrue: [^self].

	self error: function name, ' failed, response code ', 

		responseCode printString



convertArgumentDictionary: arguments

	"Convert  to an OS structure."

	| answer index |

	answer := SmArg new: arguments size.

	index := -1.

	arguments keysAndValuesDo: [:key :value |

 		answer 

			at: (index := index + 1) 

			put: (SmArg new 

				name: key asPointer; 

				value: (self convertArgument: value))].

	^answer abtMoveToOSMemory 

		abtAsExternalPassedPointer



convertArgument: argument

	argument isInteger ifTrue: [^argument].

	argument isString ifTrue: [^argument asPointer asInteger].

	argument == true ifTrue: [^1].

	argument == false ifTrue: [^0].

	^argument abtMoveToOSMemory address



convertVocabulary: vocabulary

	"Convert to an OS structure."

	| answer argument word |

	answer := OSObjectPointer 

		new: vocabulary size 

		itemType: SmVocword.

	vocabulary doWithIndex: [:string :index |

		word := SmVocword new

			flags: 0;

			spellingSize: string size;

			spelling: string asPointer.

 		answer 

			at: index - 1 

			put: word abtMoveToOSMemory address].

	^answer abtMoveToOSMemory



private querying

client

	^client

client: anObject

	client := anObject

isOpen

	^isOpen

resultContainer

	^resultContainer

newResultContainer

	^OSInt32 new abtMoveToOSMemory

result

	^self contentsOf: resultContainer

contentsOf: container

	^container uint32At: 0

sessionNumber

	^sessionNumber



private simple API calls

requestFocus

	self simpleCall: SmSesRequestFocus 

releaseFocus

	self simpleCall: SmSesReleaseFocus 

microphoneOn

	self simpleCall: SmSesMicOn

microphoneOff

	self simpleCall: SmSesMicOff

recognizeNextWord

	self simpleCall: SmSesRecognizeNextWord

haltRecognizer

	self simpleCall: SmSesHaltRecognizer

simpleCall: function

	self call: function

		with: self sessionNumber

		with: self resultContainer



private complex API calls

callOpen

	"Must be done to obtain a session number."

	| arguments |

	arguments := Dictionary new

		at: SmNapplicationName 

			put: 'Speech Shell';

		at: SmNwindowHandle 

			put: self osWidget handle asInteger;

		at: SmNconnectionId 

			put: self osWidget commandIdentifier;

		yourself.

	self call: SmSesOpen

		with: self resultContainer

		with: arguments size

		with: (self convertArgumentDictionary: arguments).

	sessionNumber := self result



callConnect

	"Must be done to be able to use a session number."

	| arguments |

	arguments := Dictionary new

		at: SmNrecognize put: true asInteger;

		at: SmNuserId put: SmUseCurrent;

		at: SmNtask put: SmUseCurrent;

		at: SmNenrollId put: SmUseCurrent;

		yourself.

	self call: SmSesConnect

		with: self sessionNumber

		with: arguments size

		with: (self convertArgumentDictionary: arguments)

		with: SmAsynchronous.



defineVocabulary: name words: words

	self call: SmSesDefineVocab

		with: self sessionNumber

		with: name asPointer

		with: words size

		with: (self convertVocabulary: words)

		with: self resultContainer



enableVocabulary: name

	self call: SmSesEnableVocab

		with: self sessionNumber

		with: name asPSZ

		with: self resultContainer.



private speech event handling

userMessages

	^#(opened heard: distracted closed)

engineMessages

	^##(Array

		with: SmConnectReply

		with: SmRecognizedWord

		with: SmFocusLost

		with: nil)

userEventHandlers: engineSelector

	|  index |

	index := self engineMessages indexOf: engineSelector.

	index = 0 ifTrue: [^#()].

	^handlers 

		at:  (self userMessages at: index)

		ifAbsent: [#()]



processSpeechMessage: lParam

	| message engineEvent parameters userHandlers |

	self client isNil ifTrue: [^self].

	message := self getSpeechMessage: lParam.

	engineEvent := self getEngineEventSelector: message.

	engineEvent = SmConnectReply ifTrue: [isOpen := true].

	parameters := engineEvent = SmRecognizedWord 

		ifTrue: [Array with: (self getWord: message)] 

		ifFalse: [#()].

	userHandlers := self userEventHandlers: engineEvent.

	self userDispatchHandlers: userHandlers with: parameters



getSpeechMessage: lParam

	self call: SmSesReceiveMsg 

		with: self sessionNumber 

		with: lParam 

		with: self resultContainer.

	^self result



getEngineEventSelector: message

	self call: SmGetMsgType 

		with: message 

		with: self resultContainer.

	^self result



getWord: message

	| wordsPointerContainer wordCountContainer wordsPointer |

	wordsPointerContainer := self newResultContainer.

	wordCountContainer:= self newResultContainer.

	self call: SmGetFirmWords

		with: message

		with: wordCountContainer

		with: wordsPointerContainer.

	wordsPointer := SmWord address: (self 

		contentsOf: wordsPointerContainer).

	^String copyFromOSMemory: (wordsPointer at: 0) spelling



userDispatchHandlers: messages with: parameters

	messages do: [:directedMessage |

		directedMessage arguments: parameters; send]

